A Distortion Synthesis Tutorial

Victor Lazzarini

An Grúpa Theicneolaíocht Fuaime agus Ceoil Dhigitigh

NUI Maynooth Ireland

Background



Winham and Steiglitz
Closed-form summation, 1969

Chowning, *FM Synthesis*, 1973

Moorer, *Summation Formulae*, 1975



LeBrun, *Digital Waveshaping*, 1979



Ishibashi, deceased, *Phase Distortion*, 1985

Puckette, *PAF,1995*

The Problem

How do we generate a complex time-evolving sound composed of many discrete components (ie. harmonic, or inharmonic, partials)?

The brute force approach:

Use one sinewave oscillator plus a pair of envelopes (amp, freq) per partial, then mix all the sources together.

The elegant solution:

Find a way of combining a few simple sources (ie. sinewave oscillators) to generate lots of components

The Problem, mathematically stated

$$\sum_{k=0}^{N-1} a_k \cos(\omega_k + \phi_k)$$

In other words, we want to generate a sound with N components summed (mixed) together (Σ), each component with:

amplitude a_k

frequency ω_k

phase ϕ_k

The simplest case is the Fourier series of a pulse with N harmonic components

$$\sum_{k=1}^{N} \cos(k\,\omega_0)$$

Closed-form Summation

The first of these ways to combine sinusoids to create a harmonic series is to employ a well known mathematical device: a closed-form sum. This is a well known one:

$$\sum_{k=-N}^{N} r^k = r^{-N} \frac{1-r^{2N+1}}{1-r}$$

For the harmonic series, we have a well-defined expression:

$$\frac{1}{N}\sum_{k=1}^{N}\cos(k\omega_0) = \frac{1}{2N}\left[\sum_{k=-N}^{N} e^{ik\omega_0} - 1\right] = \frac{1}{2N}\left[e^{-iN\omega_0}\frac{1-e^{i[2N+1]\omega_0}}{1-e} - 1\right]$$

$$= \frac{1}{2N}\left[\frac{e^{-[N+\frac{1}{2}]i\omega_0} - e^{[N+\frac{1}{2}]i\omega_0}}{e^{-i\frac{\omega_0}{2}} - e^{i\frac{\omega_0}{2}}} - 1\right] = \frac{1}{2N}\left[\frac{\sin\left((2N+1)\frac{\omega_0}{2}\right)}{\sin\left(\frac{\omega_0}{2}\right)} - 1\right]$$

Discrete Summation Formulae

Moorer provided more general principles for closed-form sums. In particular, he came up with several expressions that could generate a variety of spectral combinations, for instance:

$$\frac{\sin(\omega)-g\sin(\omega-\theta)}{1-2g\cos(\theta)+g^2}=\sum_0^{\infty} g^k\sin(\omega+k\theta)$$

In general, for a bandlimited spectrum, we have

$$\frac{\sin(\omega)-g\sin(\omega-\theta)-g^N\sin(\omega+[N+1]\theta)+g^{N+1}\sin(\omega+N\theta)}{1-2g\cos(\theta)+g^2}=\sum_{k=0}^{N-1} g^k\sin(\omega+k\theta)$$

Implementation Example

```
opcode NBlsum,a,kkkki

  ka,kw,kt,kk,itb xin
  aphw phasor kw
  apht phasor kt
  asin1 tablei aphw,itb,1
  asin2 tablei aphw - apht,itb,1,0,1
  acos tablei apht,itb,1,0.25,1
  ksq = kk*kk
  asig = (asin1 - kk*asin2)/(1 - 2*kk*acos + ksq)
   knorm = sqrt(1-ksq)
   xout asig*ka*knorm

endop
```

Waveshaping

Waveshaping, or non-linear mapping, is a method of producing lots of components by distorting the shape of a sinusoid with a function:

$$f(\cos(\omega))$$

if *f(x)* is non-linear (ie. not a straight line), then a number of harmonic components will be produced. For instance:

$$f(x)=\frac{1-gx}{1-gx+g^2}$$

$$f(\cos(\omega))=\frac{1-g\cos(\omega)}{1-2g\cos(\omega)+g^2}=\sum_{k=0}^{\infty} g^k\cos(k\omega)$$

Hyperbolic Tangent Waveshaping

Another distortion method that can be useful for creating classic analogue waveforms is the hyperbolic waveshaping method:

$$square(w) = \tanh(k \sin(\omega_m))$$



By limiting the amount of modulation, it is possible to produce nearly bandlimited square waves. From these it is simple to produce sawtooths:

$$saw(\omega) = square(\omega)(\cos(\omega)+1)$$

Implementation Example

```
opcode Waveshape,a,kkkiii
   kamp,kf,kndx,isin,itf,igf  xin
   asin oscili 0.5*kndx,kf,isin
   awsh tablei asin,itf,1,0.5
   kscl tablei kndx,igf,1
       xout awsh*kamp*kscl
endop

opcode Sawtooth,a,kkkiii
 kamp,kf,kndx,isin,itf,igf  xin
 amod    oscili  1,kf,1, 0.25
 asq       Waveshape kamp*0.5,kf,kndx,isin,itf,igf
    xout  asq*(amod + 1)
endop
```

```
; function tables:
f2 0 16385 "tanh" -157
157
f3 0  8193   4   2   1
```

Asymmetrical FM

Palamin et al built on previous work by Moorer to propose this interesting formula for changing the symmetry of the double-sided FM spectrum

$$s(t)=\exp(0.5\mathrm{k}(r-\frac{1}{r})\cos(\omega_m))\sin(\omega_c+0.5\mathrm{k}(r+\frac{1}{r})\sin(\omega_m))$$

$$=\sum_{n=-\infty}^{\infty} r^n j_n(k)\sin(\omega_c+n\,\omega_m)$$

As the expanded sum demonstrates, this includes a new scaling variable *r*, that will shift the symmetry of the spectrum away from the carrier frequency.

Implementation Example

```
opcode Asfm,a,kkkkkii

kamp,kfc,kfm,knx,kR,ifn,ifn2 xin

kndx =  knx*(kR+1/kR)*0.5
kndx2  = knx*(kR-1/kR)*0.5

afm oscili  kndx/(2*$M_PI),kfm,ifn
aph phasor  kfc
afc  tablei  aph+afm,ifn,1,0,1

amod oscili kndx2, kfm, ifn, 0.25
aexp  tablei (abs(kndx2) - amod)/50, ifn2, 1
   xout kamp*afc*aexp

endop
```

The exponential is implemented with a function table lookup. In order to use the table lookup limiting mechanism, we draw up an inverse exponential from 0 to -50 and then reverse the sign of the argument to it.

```
;(inverse) exponential function table
f5 0 131072 "exp" 0 -50 1
```

Phase Aligned Formant Synthesis

Puckette, in 95, proposed PAF as ring-modulation of a sinusoid with a complex wave. This is actually similar to one of Moorer's double-sided DSF equations

$$\sum_{k=-\infty}^{\infty} g^{|k|} \cos(\omega_c + k\,\omega_0) = \cos(\omega_c)[1 + 1\sum_{k=1}^{\infty} g^k \cos(k\,\omega_0)]$$

$$= \cos(\omega_c)[\frac{1-g^2}{1-2g\cos(\omega_0)+g^2}] = c(\omega_c) M(\omega_0)$$

The complex modulating wave can be implemented using waveshaping, so this is a distortion technique. For this purpose the formula for *M(.)* above is rearranged into:

$$M(\omega) = \frac{1+g}{1-g} f(2\sqrt{g}\frac{\sin(\omega/2)}{1-g}), \text{ with } f(x) = \frac{1}{1+x^2}$$

Implementation Example

```
opcode PAF,a,kkkkki

 kamp,kfo,kfc,kfsh,kbw,itb xin
 kn = int(kfc/kfo)
 ka = (kfc - kfsh - kn*kfo)/kfo
 kg = exp(-kfo/kbw)
 afsh phasor  kfsh
 aphs phasor  kfo/2
 a1  tablei  2*aphs*kn+afsh,1,1,0.25,1
 a2  tablei  2*aphs*(kn+1)+afsh,1,1,0.25,1
 asin  tablei  aphs, 1, 1, 0, 1
 amod Func 2*sqrt(kg)*asin/(1-kg)
 kscl = (1+kg)/(1-kg)
 acar = ka*a2+(1-ka)*a1
 asig = kscl*amod*acar
     xout asig*kamp

endop
```

```
opcode Func,a,a

asig xin
    xout  1/(1+asig^2)

endop
```

Modified FM Synthesis

The modified FM synthesis method is a variation on classic FM, which exhibits *modified* Bessel Functions in its expansion:

$$e^{k\cos(\omega_m)-k}\cos(\omega_c)=e^{-k}\sum_{n=-\infty}^{\infty} I_n(k)\cos(\omega_c+n\,\omega_m)$$

Its advantage is that *modified* Bessels are unipolar and decreasing, which allows for a better-behaved spectrum, with similar computational costs to FM (ie. very little).

Applications for bandlimited classic analogue waveform generation, among other uses, have been proposed.

Implementation Example

```
opcode ModFM,a,kkkkii

kamp,kfc,kfm,kndx,isin,iexp  xin
acar oscili kamp,kfc,isin,0.25
acos oscili 1,kfm,isin,0.25
amod table -kndx*(acos-1)/50,iexp,1
     xout acar*amod

endop
```

```
;(inverse) exponential function
table
f5 0 131072 "exp" 0 -50 1
```

Conclusion

Distortion techniques provide efficient and elegant ways of synthesising sounds. With the advent of other methods, research interest in these had vanished for a number of years. Some novel formulations and applications have rekindled interest in the area.

Although the basic techniques flourished in the brave days of early computer music, there are a number of interesting possibilities still left to be pursued.